
INTRODUCTION 
 

About 
Dynamic 
Array 
Formulas in 
Excel 

 

 Summary 
• Dynamic Arrays are the biggest change to Excel formulas in years. 

Maybe the biggest change ever. This is because Dynamic Arrays let 

you easily work with multiple values at the same time in a formula. 

• An array formula is a formula th`at can perform multiple 

calculations on one or more items in an array. Array formulas can 

return either multiple results, or a single result. 

• The key benefit of Dynamic Arrays is the ability to work with 

multiple values at the same time in a formula. This is a big upgrade 

and a welcome change. Dynamic Arrays solve some really hard 

problems in Excel, and will fundamentally change the way 

worksheets are designed. Once you see how they work, you'll 

never want to go back. 

Availability 
• Dynamic arrays and the new functions below are only available 

Excel 365, Excel 2024, Excel 2021.  

Excel 2019 and earlier do not offer dynamic array formulas. 

New Functions 
• As part of the dynamic array update, Excel now includes 8 new 

functions that directly leverage dynamic arrays to solve problems 

that are traditionally hard to solve with conventional formulas. 

Eg. FILTER, RANDARRAY, SEQUENCE, SORT, SORTBY, UNIQUE, 

XLOOKUP, XMATCH 

• As of September 2024, almost 50 new functions have been added 

to Excel 

The power of one 
• One of the most powerful benefits of the "one formula, many 

values" approach is less reliance on absolute or mixed references. 

As a dynamic array formula spills results onto the worksheet, 

references remain unchanged, but the formula generates correct 

results. 

• It is important to understand that dynamic array behavior is native 

and deeply integrated. When any formula returns multiple results, 

these results will spill into multiple cells on the worksheet. This 

includes older functions not originally designed to work with 

dynamic arrays. 

New and old array formulas 

• In Dynamic Excel, there is no need to enter array formulas with 

CTRL + SHIFT + ENTER (CSE). When a formula is created, Excel 

checks if the formula might return multiple values. If so, it will 

automatically be saved as a dynamic array formula, but you will not 

see curly braces. 

• If you open the same formula in Legacy Excel, you'll see curly 

braces.  

• If a "traditional" array formula is opened in Dynamic Excel, you will 

see the curly braces in the formula bar. However, if you re-enter 

the formula with no changes, the curly braces are removed, and 

the formula returns the same result.Spilling limitations and the 

plus (+) operator 

• The term "lifting" refers to an array calculation behavior in 

Excel formulas. When you give a range or array to a function not 

programmed to accept arrays natively (i.e. an older function), 

Excel will "lift" the function and call it multiple times, one time 

for each value in the array. The result is an array with the same 

dimensions as the input array. Lifting is a built-in behavior that 

happens automatically. 

• In Dynamic Excel, some older functions like EOMONTH "resist" 

spilling when provided a range. So for example 

EOMONTH(A1:A5,1) will return #VALUE even with valid dates in 

A1:A5. This limitation comes from certain functions expecting a 

single value instead of a range. The #VALUE! error is essentially 

reporting the range as an unexpected value. However, adding 

an operator in front of the reference will often fix the problem.  

• For example, EOMONTH(+A1:A5,1) will work and spill properly. 

This is because adding the + in front of A1:A5 forces Excel to 

evaluate the expression first, before the function runs. The 

result of +A1:A5 is an array containing 5 values, which are then 

passed into EOMONTH as the start date. EOMONTH returns 5 

results through the normal process of lifting, which is not 

function-specific.  

• Other functions that have this same limitation include EDATE, 

ISEVEN, ISODD, YEARFRAC, WORKDAY, and WORKDAY.INTL. 

  The @ character 
• With the introduction of dynamic arrays, you're going to see the 

@ character appear more often in formulas. The @ character 

enables a behavior known as "implicit intersection". Implicit 

intersection is a logical process where many values are reduced 

to one value.  

• In Legacy Excel, implicit intersection is a silent behavior used 

(when necessary) to reduce multiple values to a single result in 

one cell. In Dynamic Excel, it is not typically needed, since 

multiple results can spill onto the worksheet. When it is 

needed, implicit intersection is invoked manually with the @ 

character. 

• When opening spreadsheets created in an older version of 

Excel, you may see the @ character added automatically to 

existing formulas that have the potential to return many values. 

In Legacy Excel, a formula that returns multiple values won't 

spill on the worksheet. The @ character forces this same 

behavior in Dynamic Excel so that the formula behaves the 

same way and returns the same result as it did in the original 

Excel version. 

• In other words, the @ is added to prevent an older formula 

from spilling multiple results onto the worksheet. Depending 

on the formula, you may be able to remove the @ character and 

the behavior of the formula will not change. 

DYNAMIC ARRAYS & SPILL 
 

New 
Dynamic 
Array 
Functions 

 

 1. FILTER 
2. SORT 
3. SORTBY 
4. UNIQUE 
5. SEQUENCE 
6. RANDARRAY 
7. @ (added for backward compatibility) 

   

Important 
Terms 

 

 Legacy Excel: This is the Excel version prior to Dynamic-Array aware Excel. 
 
Dynamic Array: Any formula that returns multiple results is a Dynamic 
Array Formula. 
 
Spill Range: If a Dynamic Array formula spills the answers to multiple cells, 
then we have a “SPILL” Range. 
 
# Sign: Hash sign/ Hashtag. 
 
CTRL + SHIFT + ENTER (CSE) Formulas: Array formulas used by more 
advanced Excel users that requires pressing CSE instead of ENTER when 
using legacy Excel. 

   

Dynamic 
array 
formulas 
and spilled 
array 
behavior 

 

 Excel formulas that return a set of values, also known as an array, return 
these values to neighboring cells. This behavior is called spilling. 
 

• When you press Enter to confirm your formula, Excel will 

dynamically size the output range for you, and place the results 

into each cell within that range 

• Only the first cell in the spill area is editable 

• Cannot copy the value in the spilled range (note: but can copy and 

paste as VALUE only) 

• Spilled range comes with shadowed blue border 

• Source formatting is not carried over 

• #SPILL! error means there is a blockage that prevents it from 

spilling 

• Excel Table as Source Data 

• Spilled array NOT supported in Excel Table, you must use it outside 

Excel Table 

• Use # to reference a spilled range 

• Legacy array formulas entered via CTRL+SHIFT+ENTER (CSE) are 

still supported for back compatibility reasons, but should no 

longer be used.  

• Excel has limited support for dynamic arrays between workbooks, 

and this scenario is only supported when both workbooks are 

open. If you close the source workbook, any linked dynamic array 

formulas will return a #REF! error when they are refreshed. 

RULES FOR DYNAMIC ARRAYS 
 

Rules for 
Dynamic 
Arrays 

 

 1. If a formula delivers a single answer, it will return a single answer 
in one cell. 

2. If a formula delivers more than one answer, it will return all 
answers in more than one cell. 
Meaning the answers “SPILL” into the next cells. 

3. The actual formula lives only in the first cell of the spilled range. 
The rest are based off the first one. 

4. If something is in a cell below the spill range, you will get the new 
#spill error (your values below will never be overwritten). Once the 
values are removed the formula will spill automatically. 

5. Referencing a spilled range is done by referencing the first cell in 
the range and then putting the # sign.  
The spilled range, will automatically expand or shrink as the 
formula input changes. 

6. Spilled Array formulas are not supported inside Excel Tables. 
You need to place them outside the table. 

7. Dynamic Array functionality is not fully supported between 
workbooks. 
The workbooks need to be opened. 

THE DOT IN DYNAMIC ARRAY FORMULA 
 

About the 
dot (.) in 
Formula 

 

 The dot (.) in dynamic array formulas is a new operator introduced in 
Microsoft 365 that refines dynamic ranges by automatically excluding 
blank or empty cells. Known as the "trim reference dot operator," it allows 
formulas to adapt more precisely to the size of the actual data, removing 
unnecessary zeros or blank results that can appear when referencing 
entire columns. 
 
Syntax: The dot is placed adjacent to the colon (:) in a range reference, 
effectively telling Excel to trim the range. 
 
Trim trailing blanks: C4:.C26 creates a dynamic range that starts at cell C4 
and ends at the last non-empty cell in column C, but no later than C26. 
 
Trim leading blanks: C4.:C26 creates a dynamic range that starts at the 
first non-empty cell in column C and ends at cell C26. 
 
Trim both: C4.:.C26 trims both leading and trailing blank cells. 
 
Dynamic and automatic: As you add or remove data within the specified 
range, any formulas using the dot operator will automatically adjust their 
size to include only the relevant data. This makes formulas more robust 
and prevents errors or misleading results caused by empty cells. 
 
Key Benefits 

• Removes trailing blanks/zeros when referencing columns. 

• Makes formulas more dynamic and precise. 

• Reduces need for wrapping formulas with FILTER(...,<>””) just to 

exclude blanks. 

Availability 
 
As a recent feature, the dot operator may not be available to all users. It is 
a new addition designed for the most recent versions of Excel, particularly 
Microsoft 365, and its full implementation may still be rolling out. 

NEW FUNCTIONS 
 

FILTER 
Function 
(New 
Power 
Lookup) 

 

 • To show multiple results for one or more lookup values 
• To have a filter that automatically refreshes without the need to 

press the refresh button. 
• To sum or count the filtered values (used as an alternative to 

SUMIFS or COUNTIFS)  
• FILTER Function arguments: 

 
FILTER(array, include, [if_empty]) 
 
Array: The columns that will be showed in the result; can be the 
entire data set or a part of the original data set. 
Include: Test to check what to include for the logical test 
If_Empty: If FILTER doesn’t filter anything, what would you like to 
returned. If you leave it blank, it will show #CALC! 

   

Using 
FILTER 
Function 

 

 1. Open the sample file: DYNAMIC ARRAYS BASICS 

2. Go to Sheet: FILTER 

 
 

3. Set the following formulas: 

Cell Formula 

E7 =FILTER(A6:C33,B6:B33=F5,"Not Found") 

I7 =FILTER(A6:C33,C6:C33>J5,"Not Found") 

 

Try to highlight B6:B33=F5, and press F9 

Using 
TABLE as 
Data 
Source 

 

 1. Open the sample file: DYNAMIC ARRAYS BASICS 

2. Go to Sheet: FILTER (Using Table) 

3. To convert the current range to a TABLE, click on CELL A5, 

Press CTRL + T  

OR 

Go to INSERT menu → TABLE → OK Button 

4. Select the table, go to TABLE DESIGN menu → Change the table 
name under PROPERTIES group: 

 
 

  
 

5. Set the following formulas: 
 

Cell Formula 

E7 =FILTER(TableSales,TableSales[Container]=F5,"Not 

Found") 

I7 =FILTER(TableSales,(TableSales[Segment]=J4)*(TableSa

les[Sales]>J5),"Not Found") 

 
Note: The (*) asterisk here means AND logic. 

SORT & 
SORTBY 
Functions 

 

 • To automatically sort categories alphabetically without the need 

to “refresh” a query. 

• To sort numbers in ascending or descending order together with 

their categories. 

• Use SORT if needs to sort the entire data set. 

• Use SORTBY if needs to sort by a category not in the result. 

• SORT Function arguments: 

 

SORT(array, [sort_index], [sort_order], [by_col]) 

 

Array: The data set to be sorted. 

Sort_index: The column index number to sort on. Default is 1. 

Sort_order: Default is 1 (ascending), -1 for descending. 

By_col: Default is FALSE. If the data is in columns instead of rows, 

you can switch to TRUE. 

 

• SORTBY Function arguments: 

 

SORTBY(array, by_array1, [sort_order1] …) 

 

Array: The part of the original data set that includes your result. 

By_array1: Decide the sort order of the final answer (the array). 

Sort_order1: Default is ascending, -1 for descending. This is the sort 

order to apply to by_array1. 

 

1. Open the sample file: DYNAMIC ARRAYS BASICS 

2. Go to Sheet: SORT 

 

3. Set the following formulas: 
 

Cell Formula 

E6 =SORT(A6:C33,3,-1) 

I6 =SORT(A6:C33,{2,1}) 

M6 =SORT(A6:C33,{2,3},{1,-1}) 4.  

  4. Go to sheet: SORT (BY) 

 
 

5. Convert the data to TABLE by pressing CTRL + T 

6. Assume the table name now is TABLE2 

7. Set the following formula: 

 

Cell Formula 

F6 =SORTBY(Table2[Ref]:Table2[Container], Table2[Sales], 

-1) 
 

   

UNIQUE 
Function 
(Get a 
Unique List 
of Values) 

 

 • To get a unique list of items from data set. 

• To get a distinct list of items from data set. 

• To update unique list automatically without the need to refresh a 

query. 

 

UNIQUE(array, [by_col], [occurs_once]) 

 

Array: The part of the original data set that includes in the results. 

By_col: Default is FALSE to compare by row. TRUE to compare by 

column. 

Occurs_once: Default is FALSE means the values occur one or more 

times. TRUE means the values returned occur only once in the data 

set. 

 

1. Open the sample file: DYNAMIC ARRAYS BASICS 

2. Go to Sheet: UNIQUE 

 

3. Set the following formulas: 

Cell Formula 

F6 =UNIQUE(A6:A40) 

F6 =COUNTA(F6#) 

 
4. In Cell A6, type OTHERS 

5. Then set the following formulas: 

Cell Formula 

I6 =UNIQUE(A6:A40,,TRUE) 

K7 =UNIQUE(A6:B40) 

SEQUENCE 
Function 
(Generate a 
list of 
sequential 
numbers) 

 

 • To generate a list of index numbers. 

• To simulate dates based on specific intervals. 

• Modeling and simulations. 

• To transform legacy Excel formulas to power formulas. 

• The power of SEQUENCE lies when used in combination with other 

functions. 

 

SEQUENCE(rows, [column], [start], [step]) 

 

Rows: The number of rows the result should spill to.  

Columns: Default is 1. The number of columns the result should 

spill to. 

Start: Default is 1. The starting number. 

Step: Default is 1. The number to increment (step) by. 

 

6. Open the sample file: DYNAMIC ARRAYS BASICS 

7. Go to Sheet: SEQUENCE 

 

8. Set the following formulas: 
Cell Formula 

A6 =SEQUENCE(10) 

C6 SEQUENCE(10,,TODAY()) 

=TEXT(SEQUENCE(10,,TODAY()),"dd/mm/yy") 

D6 =TEXT(C6#,"dddd") 

G6 =MID(G5,SEQUENCE(LEN(G5)),1) 

L6 =SEQUENCE(1,12) 

=TEXT(DATE(2025,SEQUENCE(1,12),1),"mmm") 

 
 

   

‘@ Prefix 
for 
Compatibili
ty in 
Formulas  

 

 • Older Excel files opened in the new Dynamic Array Excel might 

have formulas with the @ symbol as prefix. This is to ensure that 

the formula behaves as before and spiling is prevented. 

• This was first called the SINGLE function. SINGLE function will be 

placed before the original array formulas. It was changed to the @ 

symbol in February 2029 to avoid confusion and is easier to apply 

than a whole new function. 

• @ behaves like a table row reference. It avoids spillage and returns 

the value on the same row for range references – or the first result 

in an array (also referred to as implicit intersection). 

 

@(Formula or Reference)  

 

where  the value is generally a reference to many cells, or an array 

formula. 

XLOOKUP FUNCTION 
 

About 
 

 XLOOKUP and VLOOKUP are both handy tools that help find specific 
information in a dataset. XLOOKUP was introduced in 2022 and is more 
flexible, works faster, and finds data more accurately than VLOOKUP. 
Limitations of VLOOKUP 

• It takes data only from the right side of the search column and 
cannot pull data exclusively from the left of the search column. 

• VLOOKUP is not case-sensitive, treating uppercase and lowercase 
letters identically, and potentially causing mismatches. 

• Limited search mode. 
• As the dataset gets bigger, VLOOKUP may become less efficient, 

leading to slower response times 
 

XLOOKUP can overcome those limitations. XLOOKUP easily handles two-
way searches and manages large data without any difficulty. It simplifies 
complex tasks, making your work hassle-free. 
 
=XLOOKUP(lookup_value, lookup_array, result_array, [if_not_found], 
[match_mode], [search_mode]) 
 
Lookup_value: The value you want to search for in the lookup_range. 
lookup_array: The range where the search_key will be located. 
Result_array: The range where the related result is found. 
If_not_found: Specifies the value or action if no match is found. 
Match_mode: Default is 0 – EXACT MATCH. Determines the match type - 
exact match, partial match, etc. 
Search_mode: Default is 1 – Search FIRST TO LAST. Defines how to handle 
different types of matches. 
 
Advantages of using XLOOKUP: 

• Bidirectional lookup: XLOOKUP can search both horizontally and 

vertically within a table, offering greater flexibility in searching. 

• Handling errors: It handles errors more effectively than VLOOKUP, 

providing an alternative value or message when a lookup fails. 

• Multiple criteria: It supports multiple criteria, enabling more 

complex search operations than VLOOKUP. 

• Flexibility: Returns an exact match, the next smaller item, or the 

next larger item when an exact match is not found. 

   

Use 
Wildcard 
Characters 
in 
Searches 

 

 Use wildcard characters as comparison criteria for text filters and when 
content searching and replacing. 
 

Symbol To find 

? Any single character. 

Eg. Te?t finds “Test” and “Text” 

* Any number of characters. 

Eg. *east finds “Northeast” and “Southeast” 

~ followed 

by ?, *, or ~ 

A question mark, asterisk, or tilde. 

Eg. Sure~? finds “Sure?” 

Using 
XLOOKUP 

 

 1. Open the sample file: DYNAMIC_ARRAYS_XLOOKUP 

2. Go to Sheet: SUMMARY 

 

 
 

3. Set the following formulas/ actions: 

 

Cell Formula / Action 

B8 Set data validation for selection of Jan to Dec 

 
B5 =XLOOKUP(B3,Data!B:B,Data!A:A) 

=XLOOKUP(B3,Data!B:B,Data!A:A,,2) 

B6 =XLOOKUP(B3,Data!B:B,Data!O:O) 

=XLOOKUP(B3,Data!B:B,Data!O:O,,2) 

B9 =XLOOKUP(B3,Data!B6:B22,XLOOKUP(Summary!B8,Da

ta!C5:N5,Data!C6:N22)) 

=XLOOKUP(B3,Data!B6:B22,XLOOKUP(Summary!B8,Da

ta!C5:N5,Data!C6:N22),,2) 

Using 
XLOOKUP 
with TABLE 

 

 1. Open the sample file: DYNAMIC_ARRAYS_XLOOKUP 

2. Go to Sheet: DATATABLE 

 

 
 

3. Select the dataset, press CTRL + T to convert to TABLE FORMAT. 

Let’s say the table name is TABLE1 

4. Go to Sheet: SUMMARY TABLE 

 

 
 

5. Set the following formulas/actions: 

 

Cell Formula / Action 

B8 Set data validation for selection of Jan to Dec 

=INDIRECT("Table1[[#Headers],[Jan]:[Dec]]") 

B5 =XLOOKUP(B3,Table1[Item],Table1[Item ID]) 

=XLOOKUP(B3,Table1[Item],Table1[Item ID],,2) 

B6 =XLOOKUP(B3,Table1[Item],Table1[Total]) 

=XLOOKUP(B3,Table1[Item],Table1[Total],,2) 

B9 =XLOOKUP(B3,Table1[Item],XLOOKUP(B8,Table1[#He

aders],Table1[#Data])) 

=XLOOKUP(B3,Table1[Item],XLOOKUP(B8,Table1[#He

aders],Table1[#Data]),,2) 

CELL REFERENCES IN DYNAMIC ARRAY FORMULAS  
 

Intro 
 

 Dynamic Arrays simplify complex data tasks, making Excel faster, smarter, 
and more efficient by generating ranges (i.e., arrays) of values/data, rather 
than individual return values like traditional Excel formulas. 
 
Dynamic Arrays are resizable arrays that calculate automatically and 
return values into multiple cells based on a formula entered in a single 
cell. 
 
Through over 30 years of history, Excel has undergone many changes, but 
one thing remained constant - one formula, one cell. Even with traditional 
array formulas, it was necessary to enter a formula into each cell where 
you want a result to appear. With dynamic arrays, this rule is no longer 
true.  
 
Now, any formula that returns an array of values automatically spills into 
neighboring cells, without you having to press CTRL + SHIFT + ENTER or do 
any other moves. In other words, operating dynamic arrays becomes as 
easy as working with a single cell. 
 

1. Open file: DYNAMIC_ARRAYS_SAMPLE 

   

Example 1 
 

 1. Go to Sheet: SAMPLE1 

 

 
 

Cell Formula / Action 

E4 =A4:C31 

Example 2 
 

 1. Go to Sheet: SAMPLE2 

 

 
 

Cell Formula / Action 

D6 =C6:C33*B3 
 

   

Example 3 
 

 1. Go to Sheet: SAMPLE3 

 

 
 

Cell Formula / Action 

D5 =C5:C32/D4:J4 

Example 4 
 

 1. Go to Sheet: SAMPLE4 

 

 
 

Cell Formula / Action 

A6 =Sample1!C4:C31 

D6 =SUM(A6#) 

D7 =AVERAGE(A6#) 

D8 =COUNT(A6#) 

D9 =MAX(A6#) 

D10 =MIN(A6#) 
 

   

Example 5 
 

 1. Go to Sheet: SAMPLE5 

 

 
 

2. Select the dataset (A3:E31), go to FORMULAS menu → 

 → Select TOP ROW → OK Button. 

 
 

Cell Formula / Action 

F4 =East+West+Central 

WORKING DYNAMIC ARRAY WITH OTHER FUNCTIONS  
 

Return 
More 
Values 
from 
VLookup 

 

 1. Open file: DYNAMIC_ARRAYS_WITH_OTHER FUNCTIONS 

2. Go to sheet: VLOOKUP 

 
 

3. Set the following formula: 

Cell Formula / Action 

G5 =VLOOKUP(G4,A5:D21,{2,4},0) 

G6 =VLOOKUP(G4,A5:D21,{1,2,3,4},0) 
 

   

More 
Criteria 
Matching 
from 
SUMIFS 

 

 1. Open file: DYNAMIC_ARRAYS_WITH_OTHER FUNCTIONS 

2. Go to sheet: SUMIFS 

 
 

3. Convert dataset to a TABLE named it as TABLE1. 

4. Create Data Validation in G5: 

 
 

5. Set the following formula: 

Cell Formula / Action 

H5 =SUM(SUMIFS(Table1[Sales],Table1[Item],$G5,Table1[

Month],H1:H3)) 

 

6. Copy H5 formula, highlight I5:K5, right click the selection and 

select PASTE OPTION: FORMULA 

Using # in 
INDEX & 
MATCH 

 

 1. Open file: DYNAMIC_ARRAYS_WITH_OTHER FUNCTIONS 

2. Go to sheet: INDEX & MATCH 

 
 

3. Set the following formulas: 

Cell Formula / Action 

G6 =UNIQUE(D6:D1400) 

H5 =TRANSPOSE(UNIQUE(A6:A1400)) 

H6 =SUMIFS(E6:E1400,D6:D1400,G6#,A6:A1400,H5#) 

H1 Bookcases 

H2 Mar 

H3 =INDEX(H6#,MATCH(H1,G6#,0),MATCH(H2,H5#,0)) 
 

   

Exercise 
 

 1. Go to sheet: INDEX & MATCH (TABLE) 

2. Convert the dataset to TABLE and name it as TABLE2 

3. Set the relevant formulas as above. 

   

Combining 
FILTER, 
SUMIFS, 
VLOOKUP, 
IF 

 

 1. Open file: DYNAMIC_ARRAYS_WITH_OTHER FUNCTIONS 

2. Go to sheet: FILTER 

 
 

3. Create SHIP MODE List in F5 using Data Validation. 

4. Set the following formulas: 

Cell Formula / Action 

F8 =SORT(UNIQUE(FILTER(C5:C1399,B5:B1399=F5))) 

G8 =SUMIFS(D5:D1399,C5:C1399,F8#,B5:B1399,F5) 

H8 =G8#*VLOOKUP(G8#,K8:M13,3,1) 

I8 =IF(G8#>=AVERAGE(G8#),"Good","") 

L18 =VLOOKUP(L17,F8#:I8#,2,0) 

EXCEL CHART WITH DYNAMIC ARRAY 
 

Set 
Dynamic 
Array in 
Data Series 

 

 1. Open file: DYNAMIC_ARRAYS_WITH_OTHER FUNCTIONS 

2. Go to sheet: FILTER 

(Continue from the previous settings) 

 
 

3. In cell F5, select DELIVERY TRUCK. 

4. Highlight F7:G14, insert a COLUMN CHART. 

5. Click the CHART TITLE and link it to Cell F5. 

 
 

6. To create dynamic data series, apply naming technique for the 

following cells: 

Cell Formula / Action 

F8 From FORMULAS menu → NAME MANAGER → NEW 

Button, create name Col_Item 

 
G8 Repeat as above and name it as Col_Sales 

 7. Right-click the COLUMN CHART, select SELECT DATA option: 

 
 

8. Click EDIT (a) Button and set the SERIES VALUES: 

<sheetname>!Col_Sales 

 
9. Click OK button. 

10. Click EDIT (b) Button and set the AXIS LABEL RANG: 

<sheetname>!Col_Item 

 
11. Click OK button. 

Now, the data series changes according to the SHIP MODE 

selection. 

Create 
Simple 
Excel 
Dashboard 
with 
Controls 

 

 1. Open file: DYNAMIC_ARRAYS_WITH_OTHER FUNCTIONS 

2. Go to sheet: SIMPLE DASHBOARD 

 
 

3. Turn on DEVELOPER Ribbon. 

4. Create 3 Option Button as below: 

 
5. Right-click one of the OPTION BUTTONS → FORMAT CONTROL → 

CELL LINK: $G$3 

 
 

6. Set the following formula: 

Cell Formula / Action 

F5 =UNIQUE(FILTER(C5:C1399,B5:B1399=SWITCH(G3,1,"R

egular Air",2,"Express Air",3,"Delivery Truck"))) 

G5 =SUMIFS(D5:D1399,C5:C1399,F5#,B5:B1399,SWITCH(G

3,1,"Regular Air",2,"Express Air",3,"Delivery Truck")) 

I4 =SORT(FILTER(F:G,F:F<>"",""),2,1) 

 

7. Highlight I4:J12 (Current data set). 

8. Go to INSERT menu → CLUSTERED BAR. 

9. Right-click Y-AXIS → FORMAT AXIS → Check CATEGORIES IN 

REVERSE ORDER  

 10. Select SHIP MODE and the following chart will be updated: 

MORE ABOUT FILTER FUNCTION 
 

Boolean 
Logic is 
Important 
When 
Using 
FILTER 

 

 • TRUE in Excel takes many forms.  

• It can be 1 or any non-zero number. It can be POSITIVE or 

NEGATIVE number. 

• FALSE in Excel is a 0 or empty cell. 

• Do not put a quotation mark on Boolean e.g.  “TRUE”, “FALSE” 

• Text form is not TRUE or FALSE. 

• Math Operations on TRUE FALSE Values: 

(Use IF to test the logic, eg. Type a value in cell A1, then type the 

following formula in cell  A2: =IF(A1, “T”, “F”) 

 

Operator Formula / Action 

Multiplication 

(*) 

TRUE * TRUE = 1 

TRUE * FALSE = 0 

FALSE * FALSE = 0 

TRUE * 1 = 1 

FALSE * 1 = 0 

TRUE * 123 = 123 

FALSE * 123 = 0 

TRUE * <blank cell> = 0 

FALSE * <blank cell> =0 

TRUE * <text> = #VALUE! 

FALSE * <text> = #VALUE! 

 

Addition (+) TRUE + TRUE = 2 

TRUE + FALSE = 1 

FALSE + FALSE = 0 

TRUE + 1 = 2 

FALSE + 1 = 1 

TRUE + 123 = 124 

FALSE + 123 = 123 

TRUE + <blank cell> = 1 

FALSE + <blank cell> =0 

TRUE + <text> = #VALUE! 

FALSE + <text> = #VALUE! 

 

Subtraction (-) TRUE - TRUE = 0 

TRUE - FALSE = 1 

FALSE - FALSE = 0 

TRUE - 1 = 0 

FALSE - 1 = -1 

TRUE - 123 = -122 

FALSE - 123 = -123 

TRUE - <blank cell> = 1 

FALSE - <blank cell> = 0 

TRUE - <text> = #VALUE! 

FALSE - <text> = #VALUE! 

Exclude 
Blank, 
Zero-Value 
Cells or 
Text 

 

 1. Open file: DYNAMIC_ARRAYS_WITH_FILTER 

2. Go to sheet: EXCLUDE BLANK OR ZERO

 
 

3. Set the following formula: 

Cell Formula / Action 

D6 =FILTER(A6:B22,B6:B22<>0) 

• Highlight B6:B22 and press F9 Key to view the 

data in the array 

• Highlight B6:B22<>0 and press F9 Key to view 

the returned logic 

G6 =FILTER(A6:B22,B6:B22) 

B7 Type a word: PENDING 

 

4. You will notice that the second FILTER is not working as TEXT is 

not taken as TRUE or FALSE by Excel. 

Applying 
AND Logic 

 

 1. Open file: DYNAMIC_ARRAYS_WITH_FILTER 

2. Go to sheet: AND LOGIC

 
 

3. Set the following formula: 

Cell Formula / Action 

G6 Filter for Warehouse(West), Sales>0 
 
=FILTER(A6:E22,(C6:C22="west")*(ISNUMBER(E6:E22)*(
E6:E22>0))) 
 

M6 Filter data based on the value in cell N2 & N3 

 
 

=FILTER(A6:E22,(C6:C22=N2)*(D6:D22=N3)*(ISNUMBER

(E6:E22)*(E6:E22>0)),"No Data") 

 

S6 Assume ALL if no value in N2 or N3 

 

=FILTER(A6:E22,(IF(T2="",1,C6:C22=T2))*(IF(T3="",1,D6:

D22=T3))*(ISNUMBER(E6:E22)*(E6:E22>0)),"No Data") 

 
 

   

Applying 
OR Logic 

 

 1. Open file: DYNAMIC_ARRAYS_WITH_FILTER 

2. Make a copy of Sheet AND LOGIC and name it as OR Logic 

3. Relace all (*) with (+) 

   

Using 
Subtraction  

 

 1. Open file: DYNAMIC_ARRAYS_WITH_FILTER  

2. Go to sheet: USING SUBTRACTION 

3. Set the following formula: 

Cell Formula / Action 

G6 =FILTER(A6:E22,(C6:C22<>"west")-

(D6:D22="wmart"),"No Data") 

 

Try to understand and interpret the output. 

WORKING WITH DATA VALIDATION 
 

Searchable 
Drop-Down 

 

 4. Open file: DYNAMIC_ARRAYS_WITH_DATA VALIDATION 

5. Go to sheet: SEARCHABLE DROP-DOWN 

 
 

6. In cell D5, set the data validation for Item. 

 

Note: It only filters for any word that starting with the text you type in the 
cell.  

Using 
TABLE in 
Data 
Validation 

 

 1. Open file: DYNAMIC_ARRAYS_WITH_DATA VALIDATION 

2. Go to sheet: CELL REFERENCING 

 
 

3. Highlight cell A4:A21 and press CTRL + T, named it as TABLE1 

4. Set Data Validation in cell D5, 

SOURCE → =INDIRECT("Table1[#Data]") 

 
 

5. Highlight cell F4:F21 and press CTRL + T, named it as TABLE2 

6. Go to FORMULAS menu → NAME MANAGER → NEW button → Set 

the following name: 

 
7. Click OK button. 

8. Set Data Validation in cell D5, 

SOURCE → =TABLEITEM 

Example 1: 
Dependent 
Drop Down 
List in Data 
Validation 

 

 1. Open file: DYNAMIC_ARRAYS_WITH_DATA VALIDATION 

2. Go to sheet: DEPENDENT DROP DOWN (1) 

 
 

3. In cell D5, set the data validation for Supplier. 

4. Set the following formula: 

Cell Formula / Action 

D9 =FILTER(A5:A21,B5:B21=D5) 
 

   

Example 2: 
Dependent 
Drop Down 
List in Data 
Validation 

 

 1. Open file: DYNAMIC_ARRAYS_WITH_DATA VALIDATION 

2. Go to sheet: DEPENDENT DROP DOWN (2) 

 
 

3. Highlight SUPPLIER column and set data validation. 

4. Set the following formula: 

Cell Formula / Action 

C5:C15 =XLOOKUP(B5,$G$4:$I$4,$G$5:$I$12) 

D5:D15 =XLOOKUP(C5,$G$15:$G$31,$H$15:$J$32) 

TRANSPOSE & FREQUENCY FUNCTION 
 

Intro 
 

 • Both are great for report creation and data analysis. 

• You need to use CTRL + SHIFT + ENTER in Legacy Excel, now is 

much easier in Dynamic Array. 

   

Using 
Transpose 
Function 

 

 1. Open file: DYNAMIC_ARRAYS_WITH_TRANSPOSE & FREQUENCY 

2. Go to sheet: TRANSPOSE 

3. Convert dataset A24:B41 to TABLE format, named it as TABLE1  

 
 

4. Set the following formulas: 

Cell Formula / Action 

D5 =TRANSPOSE(A4:B21) 

D25 =TRANSPOSE(TABLE1[#All]) 
 

   

Using 
Frequency 
Function 

 

 1. Go to Sheet: Frequency 

 
 

2. Set the following formulas: 

Cell Formula / Action 

F6 =FREQUENCY(B5:B21,E6:E9) 

ARRAY SYNTAX 
 

Array 
Constants 
& Array 
Syntax 

 

 Rules for inputting array constants in new Dynamic Array Excel: 
 

• Wrap array constants in curly brackets { } 

• For Legacy Excel, select enough number of cells, type the item in { 

} and press CTRL + SHIFT + ENTER 

 

• Comma (,) represents column split 

Eg. ={1,2,3} 

 ={"Coffee","Tea","Milk"} 

 

• Semicolon (;) represents row split 

Eg. ={1;2;3} 

 ={"Coffee";"Tea";"Milk"} 

 

• Using Comma & Semicolon 

Eg. ={1,"Coffee";2,"Tea";3,"Milk"} 

   

Debugging 
Formula 
Trick 

 

 Why Debug? 
 

• Understand why formula return wrong values 

• Write complex formulas and view results one step at a time 

Debugging Formula Trick 
 

• Press F9 key on parts of the formula in the formula bar. 

• Make sure correctly include brackets before pressing F9 Key. 

• Highlight and press CTRL + C if you want to copy the result. 

• After pressing CTRL + C, press ESC or CTRL + Z before leaving the 

formula. 

   

Array 
Syntax for 
Different 
Regional 
Settings 

 

 The following notes apply to those who use Excel with regional settings 
that are different to the US. For example, European regional settings. 
 

Description US Version Europea Version 

Decimal Separator . (dot) , (Comma) 

Column Separator , (Comma) \ (Backslash) 

Row Separator ; (Semi-colon) ; (Semi-colon) 

Named 
Array  
Constant 

 

 1. Open a blank worksheet. 

2. Type the following array in cell A1: 

 

={"Planning";"Manufacturing";"Quality Control";"Material 

Management"} 

 

Press ENTER key and the result will appear as below: 

 

 
 

3. Copy the formula. 

4. Go to FORMULAS menu → DEFINE NAME button 

5. Paste the formula and name it as PRODUCTION → OK button. 

 
 

6. Select a cell and type: 

 

=PRODUCTION 

=TRANSPOSE(PRODUCTION) 

 

=ARRAYTOTEXT(PRODUCTION,1) 

=ARRAYTOTEXT(PRODUCTION,0) 

 

7. To retrieve single value (Eg. MANUFACTURING) from the array, 

type: 

 

=INDEX(PRODUCTION,2) 

Example 1: 
Add More 
Values for 
“NOT 
FOUND” 

 

 1. Open file: DYNAMIC_ARRAYS_WITH_ARRAY CONSTANT 

2. Go to sheet: FOR NOT FOUND

 
 

3. Highlight cell A5:C22 and press CTRL + T, named it is TABLE1. 

4. Set the following formula: 

 

Cell Formula / Action 

E6 =FILTER(Table1,Table1[Total Sales]>=E3,"No Data") 

E3 Type:  3,000,000 

 

You will only see NO DATA appears in E6. 

E6 =FILTER(Table1,Table1[Total Sales]>=E3,{"No Item","No 

Supplier","No Sales"}) 

 

 
 

   

Example 2: 
Find TOP & 
BOTTOM 
Values 

 

 1. Open file: DYNAMIC_ARRAYS_WITH_ARRAY CONSTANT 

2. Go to sheet: TOP BOTTOM VALUE

 
 

3. Highlight cell A5:C22 and press CTRL + T, named it is TABLE2. 

4. Set the following formula: 

Cell Formula / Action 

E6 =LARGE(Table2[Total Sales],{1;2;3}) 

G6 =SEQUENCE(H5) 

H6 =LARGE(Table2[Total Sales],SEQUENCE(H5)) 

H2 =SUM(H6#) 

H3 =AVERAGE(H6#) 

Example 3: 
Different 
Sort Order 

 

 1. Open file: DYNAMIC_ARRAYS_WITH_ARRAY CONSTANT 

2. Go to sheet: DIFFERENT SORT ORDER 

 
 

3. Set the following formula: 

Cell Formula / Action 

F6 =SORT(A6:D22,{3,4,1}) 

 

Assumption: All in Ascending Orders  

Sort by SUPPLIER first,  

then by TOTAL SALES,  

then by ITEM ID. 

 

F6 =SORT(A6:D22,{3,4,1},{1,-1,1}) 

 

Assumption:  

Sort by SUPPLIER first in Ascending Order 

then by TOTAL SALES in Descending Order 

then by ITEM ID in Ascending Order 

 

Sort Order (1 is Ascending, -1 is Descending) 

DYNAMIC REPORTING 
 

Pivot-Style 
Report with 
Dynamic 
Array 

 

 1. Open file: DYNAMIC_ARRAYS_WITH_INTERACTIVE REPORT 

2. Go to sheet: PIVOT-STYLE (1) 

 
 

3. Convert the dataset to TABLE by pressing CTRL + T, name it as 

TABLE1. 

4. Set the following formula: 

Cell Formula / Action 

G4 Set Data Validation → LIST → Source: 

=INDIRECT(“TABLE1[Country]”) 

G6 =SORT(UNIQUE(Table1[Category])) 

J5 =TRANSPOSE(UNIQUE(Table1[Segment])) 

N5 =J5# 

R5 =J5# 

H6 =SUMIFS(Table1[Sales],Table1[Category],G6#,Table1[

State],G4) 

I6 =SUMIFS(Table1[Profit],Table1[Category],G6#,Table1[

State],G4) 

J6 =SUMIFS(Table1[Sales],Table1[Category],G6#,Table1[

Segment],J5#,Table1[State],G4) 

N6 =SUMIFS(Table1[Profit],Table1[Category],G6#,Table1[

Segment],J5#,Table1[State],G4) 

R6 =IFERROR(N6#/J6#,"") 

G6:U66 Conditional Formatting → Formula 

 



EXCEL DYNAMIC ARRAYS 

36 / 54 

R6:U22 Conditional Formatting → Icon Set 

 5.  

Dynamic 
Row & 
Column in 
Pivot-Style 
Report 

 

 1. Open file: DYNAMIC_ARRAYS_WITH_INTERACTIVE REPORT 

2. Go to sheet: PIVOT-STYLE (2) 

 
 

3. Convert the dataset to TABLE by pressing CTRL + T, name it as 

TABLE2. 

4. Set the following formula: 

Cell Formula / Action 

H1:H2 Data Validation → List → Source: =$A$5:$C$5 

H3 Data Validation → List → Source: =$D$5:$E$5 

I5 =SUMIFS(SWITCH(H3,"sales",Table2[Sales],"profit

",Table2[Profit]),INDEX(Table2,,MATCH(H2,Table

2[#Headers],0)),I6#) 

I6 =TRANSPOSE(UNIQUE(INDEX(Table2[Country]:Ta

ble2[Category],,MATCH(H2,Table2[#Headers],0)))

) 

G7 =UNIQUE(INDEX(Table2[Country]:Table2[Categor

y],,MATCH(H1,Table2[#Headers],0))) 

I7 =SUMIFS(SWITCH(H3,"sales",Table2[Sales],"profit

",Table2[Profit]),INDEX(Table2,,MATCH(H1,Table

2[#Headers],0)),G7#,INDEX(Table2,,MATCH(H2,T

able2[#Headers],0)),I6#) 

H7 =SUMIFS(SWITCH(H3,"sales",Table2[Sales],"profit

",Table2[Profit]),INDEX(Table2,,MATCH(H1,Table

2[#Headers],0)),G7#) 

H5 =SUM(I5#) 

G5:XFD5 Conditional Formatting → Set Formula → Set 

Bold, Number Format with thousand separator 

 
G6:XFD6 Conditional Formatting → Set Formula → Set 

Bold, Grey Shading 

 
G7:XFD100 Conditional Formatting → Set Formula → Set 

White Shading, Bottom border line 

Quick 
Calendar 
Preparation 

 

 1. Open file: DYNAMIC_ARRAYS_WITH_INTERACTIVE REPORT 

2. Go to sheet: CALENDAR 

 
 

3. Set the following formula: 

Cell Formula / Action 

D5 =DATE(B3,MONTH(DATEVALUE(B4&"1")),1) 

D7 =WEEKDAY(D5) 

D7 =D5-WEEKDAY(D5) 

D7 =D5-WEEKDAY(D5)+1 

D7 =SEQUENCE(6,7,D5-WEEKDAY(D5)+1) 

D7 =IF(MONTH(SEQUENCE(6,7,D5-

WEEKDAY(D5)+1))<>MONTH(D5),"",SEQUENCE(6,7,D

5-WEEKDAY(D5)+1)) 

D7:J12 Press CTRL + 1 to FORMAT CELL → Select CUSTOM in 

NUMBER Tab → Type: dd 

D5 Press CTRL + 1 to FORMAT CELL → Select CUSTOM in 

NUMBER Tab → Type: mmmm yyyy 

 

4. Adjust the column width for the calendar. 

Searchable 
Drop-Down 
Based on 
Partial Text 

 

 1. Open file: DYNAMIC_ARRAYS_WITH_INTERACTIVE REPORT 

2. Go to sheet: DATA VALIDATION 

 
 

3. Set the following formula: 

Cell Formula / Action 

H1 =SEARCH(E5,A6:A22) 

H1 =ISNUMBER(SEARCH(E5,A6:A22)) 

H1 =FILTER(A6:A22,ISNUMBER(SEARCH(E5,A6:A22)),"Not 

Found") 

E5 Set Data Validation → LIST → Type: =$H$1# 

Multiple 
Match 
Results 
Across 
Multiple 
Columns 

 

 1. Open file: DYNAMIC_ARRAYS_WITH_INTERACTIVE REPORT 

2. Go to sheet: DATA IN COLUMN 

 
 

3. Convert the dataset to TABLE and name it as TABLE3 

4. Set the following formula: 

Cell Formula / Action 

E6 =UNIQUE(Table3[Product]) 

F6 = TRANSPOSE( 

FILTER(Table3[Color],Table3[Product]=E6)) 

E17 =E6# 

F17 =TEXTJOIN(";",TRUE,TRANSPOSE(FILTER(Table3[Color],

Table3[Product]=E6))) 
 

   

Approximat
e Match in 
Filter 

 

 1. Open file: DYNAMIC_ARRAYS_WITH_INTERACTIVE REPORT 

2. Go to sheet: APPROXIMATE MATCH 

 
 

3. Convert the dataset to TABLE and name it as TABLE4 

4. Set the following formula: 

Cell Formula / Action 

E6 =FILTER(Table4,Table4[Product]="*"&F3&"*") 

E6 #N/A Error is not numeric value. Therefore you can use 

ISNUMBER to check the validity of the data searched: 

 

=FILTER(Table4,ISNUMBER(SEARCH(F3,Table4[Product]))) 

Find 
TOP/BOTT
OM N in 
Filter 

 

 1. Open file: DYNAMIC_ARRAYS_WITH_INTERACTIVE REPORT 

2. Go to sheet: TOP AND BOTTOM N 

 
 

3. Convert the dataset to TABLE and name it as TABLE5 

4. Set the following formula: 

Cell Formula / Action 

F6 SORT(FILTER(Table5,Table5[Sales]>=LARGE(Table5[Sa

les],G3)),3,-1) 

K6 =SORT(FILTER(Table5,Table5[Sales]<=SMALL(Table5[

Sales],G3)),3,-1) 

E6 =SEQUENCE(G3) 

J6 =SEQUENCE(G3,,G3,-1) 

EXCEL ARRAYS BEHAVIOR 
 

Lifting 
Array 
Behavior 

 

 • If a formula expects a scalar (one value) but is given an array, the 

function lifts the multiple values to provide an output with 

multiple results. 

 

Examples: 

 

MONTH function expects one value (Date) 

If more than one value is provided, the MONTH function will return 

many values: 

 
 

Highlight the formula MONTH(A1:A3), press F9 to debug the 

values. You will see the value in array format: {4;8;12} 

 

   

Pairwise 
Lifting 
Array 
Behavior 

 

 • When multiple arrays of the same size are provided instead of two 

scalars, they are lifted in pairs. 

 

Examples: 

 

 
Same number of values are provided.  

Excel lifts them in a pairwise fashion: 

 

1+2 ; 3+4 ; 5+6 

 

 

In Legacy Excel they could not spill on the grid – There were visible 

in the formula bar but Excel was not able to show all these values 

in one cell. It only showed one result. 

Broadcastin
g Excel 
Array 
Behavior 

 

 • When we have arrays of different sizes, we need to expand the 

smaller ones to the same size as the big one so we can do pairwise 

lifting. 

 

Example 1: 

 
 

Two ranges of different size added together, the 2 in cell B1 is 

broadcasted down so Excel can do pairwise lifting. 

B1 value will be broadcasted down and B2:B3 will take the 2 value. 

 
 

If arrays of different sizes are added together, the size of the 

resulting array is equal to the largest height and the largest width 

of the source arrays. 

 

Example 2: 

 
 

Result shown as: 

Implicit 
Intersectio
n in Legacy 
Excel 

 

 • Multi values provided to Scalar Arguments. 

 

Example: 

We type the =MONTH(A1:A3) in cell B2 instead of cell B1. 

In Legacy Excel, you will get 8 only in cell B2 

In Dynamic Array, you will get all results starting from cell B2 to B4 

 
 

Implicit intersection applied to regular formulas. 

The CTRL + SHIFT + ENTER formulas turned OFF Implicit Intersect 

in Legacy Excel. 

   

Excel CSE 
Formula in 
the Future 

 

 • Suggestion from Microsoft: 

 

CSE array formulas are only retained for compatibility reason. 

Going forward, you should use dynamic arrays formulas instead. 

   

Quiz 
 

 Question 1:  
If we type this formula in Excel =SUM({1;1;1}+1) what do we get? 

a. Error 

b. 4 

c. 6 

d. 12 

Question 2:  
Now if we type this formula in Excel =SUM({1;1;1}+{1;1}) what do we get? 

a. Error 

b. 4 

c. 6 

d. 12 

Question 3:  
What about this one =SUM({1;1;1}+{1,1})? 

a. Error 

b. 4 

c. 6 

d. 12 

Question 4: The problem with legacy formulas was implicit intersection. 
Dynamic Array formulas don't do implicit intersection unless we use the @ 
prefix. 
 

a. True 

b. False 

   

   

   



EXCEL DYNAMIC ARRAYS 

45 / 54 

MORE EXCEL ARRAYS EXAMPLE 
 

Unique 
Value from 
Non-
adjacent 
Columns 

 

 1. Open file: DYNAMIC_ARRAYS_WITH_MORE EXAMPLE 

2. Go to sheet: UNIQUE VALUE-NON-ADJACENT COLUMNS 

 

 
3. Convert the dataset to TABLE and name it as TABLE1 

4. Set the following formula: 

Cell Formula / Action 

G6 =UNIQUE(Table1[Segment]:Table1[Type]) 

J6 =UNIQUE(CHOOSE({1,2},Table1[Segment],Table1[Ship 

Mode])) 

J6 =SORT(UNIQUE(CHOOSE({1,2},Table1[Segment],Table

1[Ship Mode])),{1,2},{1,1}) 

 
 

   

Filter Only 
for Non-
Blank 
Criteria 

 

 1. Open file: DYNAMIC_ARRAYS_WITH_MORE EXAMPLE 

2. Go to sheet: NON-BLANK CRITERIA FILTERING 

 
 

3. The dataset is a TABLE and the name is MYTABLE. 

And the formula already set in cell G6: 

=FILTER(myTable,(myTable[Segment]=H2)*(myTable[Type]=H3),"N

o Records") 

4. Try to blank the SEGMENT or TYPE selection, it will display as NO 

RECORDS. 

5. Refine the formula as below: 

Cell Formula / Action 

G6 =FILTER(myTable,(ISBLANK(H2)+(myTable[Segment]=

H2))*(ISBLANK(H3)+(myTable[Type]=H3)),"No 

Records") 

 
6. Try to draft the logic behinds this formula. 

Split Text & 
Numbers 
from 1 Cell 
to Multiple 
Cell 

 

 1. Open file: DYNAMIC_ARRAYS_WITH_MORE EXAMPLE 

2. Go to sheet: SPLIT DATA 

 
 

3. Set the following formula: 

Cell Formula / Action 

D5 ="Text " &SEQUENCE(,MAX(LEN(A6:A10))) 

D6 =MID(A6:A10,SEQUENCE(,MAX(LEN(A6:A10))),1) 

B6 =TEXTJOIN("",TRUE,FILTER(D6:L6,ISNUMBER(VALUE(

D6:L6)))) 

 
4. Copy formulas to other cells. 

5. The result displayed as below: 

 

   

Choose 
Columns to 
Display in a 
FILTER 

 

 1. Open file: DYNAMIC_ARRAYS_WITH_MORE EXAMPLE 

2. Go to sheet: CHOOSE COLUMNS 

 
 

3. The dataset is a TABLE and the name is TABLE 

4. Set the following formula: 

Cell Formula / Action 

G6 =CHOOSECOLS(FILTER(Table,Table[Ship 

Mode]=H3,""),2,3,5) 

Stack 
Results 
from 
Different 
Data 
Sources / 
Tables 

 

 1. Open file: DYNAMIC_ARRAYS_WITH_MORE EXAMPLE 

2. Go to sheet: STACK DATA 

 
 

3. The dataset is a TABLE and the name is TABLE2 

4. Set the following formula: 

Cell Formula / Action 

G6 =VSTACK(Table,Table2) 

G6 =CHOOSECOLS(VSTACK(Table,Table2),2,3,4) 
 

   

Take 
Number of 
Top/Botto
m Rows; 
Number of 
Left/Right 
Columns 

 

 1. Open file: DYNAMIC_ARRAYS_WITH_MORE EXAMPLE 

2. Go to sheet: TAKE DATA 

 
 

3. The dataset is a TABLE and the name is TABLETAKE 

4. Set the following formula: 

Cell Formula / Action 

F6 =TAKE(TableTake,H3) 

K6 =TAKE(SORT(TableTake,4,-1),M3) 

P6 =CHOOSECOLS(TAKE(FILTER(TableTake,(TableTake[S

egment]=P2)*(TableTake[Type]=P3)),-1),1,4) 

Magic 1: 
GROUPBY 
Function 

 

 1. Open file: DYNAMIC_ARRAYS_WITH_MORE EXAMPLE 

2. Go to sheet: GROUPBY (1) 

 
 

3. The dataset is a TABLE and the name is TABLE3 

4. Set the following formula: 

Cell Formula / Action 

G5 =GROUPBY(Table3[[Segment]:[Type]],Table3[Sales],S

UM) 

*No header but with Total Row at Bottom 

K5 =GROUPBY(HSTACK(Table3[Segment],Table3[Ship 

Mode]),Table3[Sales],SUM) 

*For adjacent columns 

O5 =GROUPBY(HSTACK(Table3[Segment],Table3[Ship 

Mode]),Table3[Sales],HSTACK(SUM,AVERAGE,MEDIA

N,COUNT,PERCENTOF)) 

*Add more aggregated data 

W5 =GROUPBY(Table3[[#All],[Segment]:[Type]],Table3[[#

All],[Sales]],LAMBDA(x,SUM(FILTER(x,ISNUMBER(x)))),

3) 

*Include Field Names 

W:Y Set Conditional Formatting using FORMULA: 

 

=OR($W1="Total",$W1="Grand Total") 

 

And set font to BOLD, add TOP Border Line 

W5 =GROUPBY(Table3[[#All],[Segment]:[Type]],Table3[[#

All],[Sales]],SUM) 

Add 2 for the TOTAL_DEPTH to display Subtotal 

 

Try to change the Conditional Formatting formula 

rules to apply the same format style (set font to 

BOLD, add TOP Border Line) for the subtotal. 

Magic 2: 
GROUPBY 
Function 

 

 1. Open file: DYNAMIC_ARRAYS_WITH_MORE EXAMPLE 

2. Go to sheet: GROUPBY (2) 

3. The dataset is a TABLE and the name is TABLE4 

4. Set the following formula: 

Cell Formula / Action 

G6 =GROUPBY(HSTACK(YEAR(Table4[Date]),TEXT(Table4

[Date],"mmm")),Table4[Sales],SUM) 

*You will notice that the month name is not sorted 

accordingly 

G6 =GROUPBY(HSTACK(YEAR(Table4[Date]),MONTH(Ta

ble4[Date]),TEXT(Table4[Date],"mmm")),Table4[Sales

],SUM) 

*By adding more month number, it will solve 

G6 =CHOOSECOLS(GROUPBY(HSTACK(YEAR(Table4[Dat

e]),MONTH(Table4[Date]),TEXT(Table4[Date],"mmm")

),Table4[Sales],SUM),1,3,4) 

*To skip the Month Number column, use 

CHOOSECOLS Function 

 

 
 

5. Continue set the following formula: 

Cell Formula / Action 

K6 =GROUPBY(Table4[Segment],HSTACK(Table4[Sales],T

able4[Sales],Table4[Ship 

Mode]),HSTACK(SUM,AVERAGE,LAMBDA(x,COUNTA(

UNIQUE(x))))) 

*The headers might be not user-friendly. 

K6 =DROP(GROUPBY(Table4[Segment],HSTACK(Table4[S

ales],Table4[Sales],Table4[Ship 

Mode]),HSTACK(SUM,AVERAGE,LAMBDA(x,COUNTA(

UNIQUE(x))))),1) 

*Use DROP function to remove the 1st Row 

K6 Change the COUNTA to ARRAYTOTEXT Function. 

How About 
PIVOTBY 
Function 

 

 1. Open file: DYNAMIC_ARRAYS_WITH_MORE EXAMPLE 

2. Go to sheet: PIVOTBY 

3. The dataset is a TABLE and the name is pTABLE 

4. Set the following formula: 

Cell Formula / Action 

H5 =PIVOTBY(pTable[Segment],MONTH(pTable[Date]),p

Table[Sales],SUM) 

 

H5 =PIVOTBY(pTable[Segment],TEXT(pTable[Date],"mm

m"),pTable[Sales],SUM) 

*Month is sorted but not in standard order of the 

month 

H5 =PIVOTBY(pTable[Segment],1*(1&TEXT(pTable[Date],

"mmm")),pTable[Sales],SUM) 

*Convert it to a valid date format and Press CTRL + 1 

to format the header to mmm 

 
 
 Try to use HSTACK Function to cater more fields in the PIVOTBY. 

INTRODUCTION TO LAMBDA & LET FUNCTION 
 

LAMBDA 
Function:  
 
Use 
Function to 
Create 
Function 

 

 The LAMBDA function in Excel is designed to create custom functions that 
can be called by user-friendly names and reused throughout a workbook. 
Due to the fact that a LAMBDA-defined function can call other functions 
or even itself as many times as needed. This feature is called "recursion", 
and this is what makes LAMBDA so effective. 
 
Example: Create new function named: GSTPRICE to get the price including 
GST rate (8%) 
 
Type the original formula first: 

 
 
Convert to LAMBDA format: 

 
 
Copy the LAMBDA formula together with = sign, and assign name in NAME 
MANAGER: 

 
 
Now, try to use the new Function name: GSTPRICE as below: 

  Example: Create a new Function named: BMI to find the bmi of a person 
with Height and Weight provided. 
 

 
 

Cell Formula / Action 

J8 =ROUND(I8/(H8*H8),1) 

K8 =LAMBDA(h,w,ROUND(w/(h*h),1))(H8,I8) 

 Create a name for this new function: 

 
L8 =BMI(H8,I8) 

 
Try to change the Parameter name: 

• h changes to Height_m 

• w changes to Weight_kg 

   

Exercise: 
Using 
LAMBDA 

 

 Use LAMBDA to create a new function called LEAP using the following 

formula: 

LEN 
Function: 
 
Concept is 
same as 
naming 
cells, 
ranges and 
formulas in 
Name 
Manager 
 

 

 The Excel LET function allows you to assign names to calculation results 

and define variables inside a formula, so that the formula looks clearer 

and works faster. 

 

Essentially, the concept is the same as naming cells, ranges and formulas 

in the Name Manager. What makes the LET function different is that the 

declared names only exist in the scope of a given formula and nowhere 

else. 

 

Example: Commission will be given if Actual more than Budget: 

 
  

Cell Formula / Action 

C8 =IF((A8-B8)<0,0,(A8-B8)) 

D8 =LET(var,(A8-B8),IF(var<0,0,var)) 

*var to replace (A8-B8) 

ASSIGNMENT 
 

Question: 
 

 Create the following data  
 

 
 
 
Use FILTER Function and Conditional Formatting to achieve the following 
result: 

 
 

 
 
 
 

  
 


